

单片机原理实验 指导书

商丘师范学院物理与信息工程系编

目 录

实验一 P1 口实验一	1
实验二 简单 I/O 口扩展实验	4
实验三 中断实验	8
实验四 定时器实验	13
实验五 串行口实验	16
实验六 A/D 转换实验	19

实验一 P1 口实验一

一、实验目的

- (1) 学习 P1 口的使用方法
- (2) 学习延时子程序的编写和使用

二、实验仪器

微机、EL-8051-III 型单片机实验箱

三、实验原理

P1 口为准双向口，P1 口的每一位都能独立地定义为输入位或输出位。作为输入位时，必须向锁存器相应位写入“1”，该位才能作为输入。8031 中所有口锁存器在复位时均置为“1”，如果后来在口锁存器写过“0”，在需要时应写入一个“1”，使它成为一个输入。

可以用第二个实验做一下实验。先按要求编好程序并调试成功后，可将 P1 口锁存器中置“0”，此时将 P1 做输入口，会有什么结果。

再来看一下延时程序的实现。现常用的有两种方法，一是用定时器中断来实现，一是用指令循环来实现。在系统时间允许的情况下可以采用后一种方法。

本实验系统晶振为 6.144MHZ，则一个机器周期为 $12 \div 6.144\mu s$ 即 $1 \div 0.512\mu s$ 。现要写一个延时 0.1s 的程序，可大致写出如下：

```
MOV R7, #X          (1)
DEL1: MOV R6, #200  (2)
DEL2: DJNZ R6, DEL2 (3)
      DJNZ R7, DEL1 (4)
```

上面 MOV、DJNZ 指令均需两个机器周期，所以每执行一条指令需要 $1 \div 0.256\mu s$ ，现求出 X 值：

$$1 \div 0.256 + X (1 \div 0.256 + 200 \times 1 \div 0.256 + 1 \div 0.256) = 0.1 \times 10$$

指令 (1) 指令 (2) 指令 (3) 指令 (4)


所需时间 所需时间 所需时间 所需时间

$$X = (0.1 \times 10^6 - 1 \div 0.256) / (1 \div 0.256 + 200 \times 1 \div 0.256 + 1 \div 0.256) = 127D = 7FH$$


经计算得 X=127。代入上式可知实际延时时间约为 0.100215s，已经很精确了。

1、程序流程图：

(1)、循环点亮发光二极管




(2)、通过发光二极管将 P1 口的状态显示




2、电路原理图:

(1) P1 口输出实验



(2) P1 口输入实验



3、实验程序:

(1)、循环点亮发光二极管 (T1_1.ASM)

```

NAME    T1_1                ;P1 口输实验
CSEG    AT    0000H
        LJMP START
CSEG    AT    4100H
START:  MOV A,#0FEH
LOOP:   RL  A                ; 左移一位, 点亮下一个发光二极管
        MOV P1,A
        LCALL DELAY         ;延时 0.1 秒
  
```

```

        JMP    LOOP
DELAY:  MOV R1,#127          ; 延时 0.1 秒
DEL1:   MOV R2,#200
DEL2:   DJNZ  R2,DEL2
        DJNZ  R1,DEL1
        RET
        END

```

(2)、通过发光二极管将 P1 口的状态显示(T1_2.ASM)

```

NAME    T1_2                ;P1 口输入实验
OUT_PORT EQU 0CFA0H
CSEG    AT 0000H
        LJMP START
CSEG AT 4100H
START:  MOV P1,#0FFH        ;复位 P1 口为输入状态
        MOV A,P1            ;读 P1 口的状态值入累加器 A
        MOV DPTR,#OUT_PORT ;将输出口地址赋给地址指针 DPTR
        MOVX @DPTR,A        ;将累加器 A 的值赋给 DPTR 指向的地址
        JMP  START          ;继续循环监测端口 P1 的状态
        END

```

四、实验步骤

1、连接 P1.0~P1.7 接发光二极管 L1~L8，执行程序 1(T1_1.ASM)，观察发光二极管循环点亮的情况。

2、修改程序使得循环时间加倍。

3、连接 P1.0~P1.7 接平推开关 K1~K8；74LS273 的 O0~O7 接发光二极管 L1~L8；74LS273 的片选端 CS273 接 CS0，执行程序 2(T1_1.ASM)，通过设置开关 K1~K8 输入二进制数 10100011，观察发光二极管点亮的情况。

五、问题思考

1、发光二极管的亮和灭与驱动电平、数字有什么关系？

2、分析寄存器 R1、R2 对延时程序的影响。

实验二 简单 I/O 口扩展实验

一、实验目的

- (1) 学习在单片机系统中扩展简单 I/O 接口的方法
- (2) 学习数据输出程序的设计方法
- (3) 学习模拟交通灯控制的实现方法


二、实验仪器

微机、EL-8051-III 型单片机实验箱


三、实验原理

要完成本实验，首先必须了解交通路灯的亮灭规律。本实验需要用到实验箱上八个发光二极管中的六个，即红、黄、绿各两个。不妨将 L1(红)、L2(绿)、L3(黄) 做为东西方向的指示灯，将 L5(红)、L6(绿)、L7(黄) 做为南北方向的指示灯。而交通灯的亮灭规律为：初始态是两个路口的红灯全亮，之后，东西路口的绿灯亮，南北路口的红灯亮，东西方向通车，延时一段时间后，东西路口绿灯灭，黄灯开始闪烁。闪烁若干次后，东西路口红灯亮，而同时南北路口的绿灯亮，南北方向开始通车，延时一段时间后，南北路口的绿灯灭，黄灯开始闪烁。闪烁若干次后，再切换到东西路口方向，重复上述过程。各发光二极管的阳极通过保护电阻接到+5V 的电源上，阴极接到输入端上，因此使其点亮应使相应输入端为低电平。

1、电路原理图：



2、程序流程图：



3、实验程序:

```

NAME    T3                ;I/O 口扩展实验一
PORT    EQU    0CFA0H     ;片选地址 CS0
CSEG    AT    0000H
        LJMP   START
CSEG    AT    4100H
START:  MOV    A,#11H     ;两个红灯亮, 黄灯、绿灯灭
        ACALL  DISP      ;调用 273 显示单元 (以下雷同)
        ACALL  DE3S      ;延时 3 秒
LLL:    MOV    A,#12H     ;东西路口绿灯亮;南北路口红灯亮
        ACALL  DISP
        ACALL  DE10S     ;延时 10 秒
  
```

```

MOV    A,#10H    ;东西路口绿灯灭;南北路口红灯亮
ACALL  DISP
MOV    R2,#05H   ;R2 中的值为黄灯闪烁次数
TTT:   MOV    A,#14H ;东西路口黄灯亮;南北路口红灯亮
ACALL  DISP
ACALL  DE02S     ;延时 0.2 秒
MOV    A,#10H   ;东西路口黄灯灭;南北路口红灯亮
ACALL  DISP
ACALL  DE02S     ;延时 0.2 秒
DJNZ  R2,TTT    ;返回 TTT，使东西路口黄灯闪烁五次
MOV    A,#11H   ;两个红灯亮，黄灯、绿灯灭
ACALL  DISP
ACALL  DE02S     ;延时 0.2 秒
MOV    A,#21H   ;东西路口红灯亮;南北路口绿灯亮
ACALL  DISP
ACALL  DE10S     ;延时 10 秒
MOV    A,#01H   ;东西路口红灯亮;南北路口绿灯灭
ACALL  DISP
MOV    R2,#05H   ;黄灯闪烁五次
GGG:   MOV    A,#41H ;东西路口红灯亮;南北路口黄灯亮
ACALL  DISP
ACALL  DE02S     ;延时 0.2 秒
MOV    A,#01H   ;东西路口红灯亮;南北路口黄灯灭
ACALL  DISP
ACALL  DE02S     ;延时 0.2 秒
DJNZ  R2,GGG    ;返回 GGG，使南北路口;黄灯闪烁五次
MOV    A,#03H   ;两个红灯亮，黄灯、绿灯灭
ACALL  DISP
ACALL  DE02S     ;延时 0.2 秒
JMP   LLL       ;转 LLL 循环
DE10S: MOV    R5,#100 ;延时 10 秒
JMP   DE1
DE3S:  MOV    R5,#30 ;延时 3 秒
JMP   DE1
DE02S: MOV    R5,#02 ;延时 0.2 秒

```

```

DE1:  MOV    R6,#200
DE2:  MOV    R7,#126
DE3:  DJNZ   R7,DE3
      DJNZ   R6,DE2
      DJNZ   R5,DE1
      RET
DISP: MOV    DPTR,#PORT    ;273 显示单元
      CPL   A
      MOVX  @DPTR,A
      RET
      END

```

四、实验步骤

1、74LS273 的输出 O0~O7 接发光二极管 L1~L8，74LS273 的片选 CS273 接片选信号 CS0，此时 74LS273 的片选地址为 CFA0H~CFA7H 之间任选。

2、运行实验程序，观察 LED 显示情况是否与实验内容相符。

3、修改程序使得车辆通过时间加倍。

4、修改程序让黄灯闪烁次数为 10 次

实验电路图：

五、问题思考

1、分析片选信号线的作用。

2、分析 74LS273 在电路中的作用。

实验三 中断实验

一、实验目的

- (1) 学习外部中断技术的基本使用方法
- (2) 学习中断处理程序的编程方法

二、实验仪器

微机、EL-8051-III 型单片机实验箱


三、实验原理

本实验中断处理程序的应用，最主要的地方是如何保护进入中断前的状态，使得中断程序执行完毕后能回到交通灯中断前的状态。要保护的地方，除了累加器 ACC、标志寄存器 PSW 外，还要注意：一是主程序中的延时程序和中断处理程序中的延时程序不能混用，本实验给出的程序中，主程序延时用的是 R5、R6、R7，中断延时用的是 R3、R4 和新的 R5。第二，主程序中每执行一步经 74LS273 的端口输出数据的操作时，应先将所输出的数据保存到一个单元中。因为进入中断程序后也要执行往 74LS273 端口输出数据的操作，中断返回时如果没有恢复中断前 74LS273 端口锁存器的数据，则显示往往出错，回不到中断前的状态。还要注意一点，主程序中往端口输出数据操作要先保存再输出，例如有如下操作：

```
MOV A, #0F0H      (0)
MOVX @R1, A      (1)
MOV SAVE, A      (2)
```


程序如果正好执行到 (1) 时发生中断，则转入中断程序，假设中断程序返回主程序前需要执行一句 MOV A, SAVE 指令，由于主程序中没有执行 (2)，故 SAVE 中的内容实际上是前一次放入的而不是 (0) 语句中给出的 0F0H，显示出错，将 (1)、(2) 两句顺序颠倒一下则没有问题。发生中断时两方向的红灯一起亮 10 秒，然后返回中断前的状态。

1、实验电路图：




2、程序流程图：

(1) 主程序框图



(2) 中断程序框图



3、实验程序：

```

NAME    T5                                ;中断控制实验
OUTPORT EQU    0CFB0H                    ;端口地址
SAVE    EQU    55H                        ;save 保存从端口 cfa0 输出的数据
CSEG    AT    0000H
        LJM    START
CSEG AT 4003H
        LJM    INT
  
```

```

CSEG AT 4100H
START: SETB  IT0
        SETB  EX0
        SETB  EA
        MOV   A,#11H      ;置首显示码
        MOV   SAVE,A      ;保存
        ACALL DISP       ;显示输出
        ACALL DE3S       ;延时 3 秒
LLL:    MOV   A,#12H     ;东西路口绿灯亮，南北路口红灯亮
        MOV   SAVE,A
        ACALL DISP
        ACALL DE10S      ;延时 10 秒
        MOV   A,#10H     ;东西路口绿灯灭
        MOV   SAVE,A
        ACALL DISP
        MOV   R2,#05H    ;东西路口黄灯闪烁 5 次
TTT:    MOV   A,#14H
        MOV   SAVE,A
        ACALL DISP
        ACALL DE02S
        MOV   A,#10H
        MOV   SAVE,A
        ACALL DISP
        ACALL DE02S
        DJNZ  R2,TTT
        MOV   A,#11H     ;红灯全亮
        MOV   SAVE,A
        ACALL DISP
        ACALL DE02S      ;延时 0.2 秒
        MOV   A,#21H     ;东西路口红灯亮，南北路口绿灯亮
        MOV   SAVE,A
        ACALL DISP
        ACALL DE10S      ;延时 10 秒
        MOV   A,#01H     ;南北路口绿灯灭
        MOV   SAVE,A

```

```

        ACALL  DISP
        MOV   R2,#05H      ;南北路口黄灯闪烁 5 次
GGG:    MOV   A,#41H
        MOV   SAVE,A
        ACALL DISP
        ACALL DE02S
        MOV   A,#01H
        MOV   SAVE,A
        ACALL DISP
        ACALL DE02S
        DJNZ  R2,GGG
        MOV   A,#11H      ;红灯全亮
        MOV   SAVE,A
        ACALL DISP
        ACALL DE02S      ;延时 0.2 秒
        JMP   LLL        ;转 LLL 循环
DE10S: MOV   R5,#100     ;延时 10 秒
        JMP   DE1
DE3S:   MOV   R5,#30     ;延时 3 秒
        JMP   DE1
DE02S:  MOV   R5,#02     ;延时 0.2 秒
DE1:    MOV   R6,#200
DE2:    MOV   R7,#126
DE3:    DJNZ  R7,DE3
        DJNZ  R6,DE2
        DJNZ  R5,DE1
        RET
INT:    CLR   EA
        PUSH  ACC        ;中断处理
        PUSH  PSW
        MOV   A,R5
        PUSH  ACC
        MOV   A,#11H     ;红灯全亮，绿、黄灯全灭
        ACALL DISP
DEL10S: MOV   R3,#100    ;延时 10 秒

```

```

DEL1:  MOV    R2,#200
DEL2:  MOV    R5,#126
DEL3:  DJNZ   R5,DEL3
      DJNZ   R4,DEL2
      DJNZ   R3,DEL1
      MOV    A,SAVE    ;取 SAVE 中保存数据输出到 cfa0 端口
      ACALL DISP
      POP    ACC        ;出栈
      MOV    R5,A
      POP    PSW
      POP    ACC
      SETB  EA        ;允许外部中断
      RETI
DISP:  MOV    DPTR,#OUTPORT
      CPL   A
      MOVX  @DPTR,A
      RET
      END

```

四、实验步骤

- 1、74LS273 的输出 O0~O7 接发光二极管 L1~L8，74LS273 的片选 CS273 接片选信号 CS2，此时 74LS273 的片选地址为 CFB0H~CFB7H 之间任选。单脉冲输出端 P-接 CPU 板上的 INT0。
- 2、按单脉冲开关，观察到红绿灯的情况。
- 3、修改程序，使得急救车通过时间加倍。

五、问题思考

- 1、CPU 板上的 INT0 端子为什么接到单脉冲输出端 P-?
- 2、74LS273 的片选 CS273 接片选信号 CS0，应怎样修改程序?

实验四 定时器实验

一、实验目的

- (1) 学习 8031 内部计数器的使用和编程方法
- (2) 进一步掌握中断处理程序的编写方法

二、实验仪器

微机、EL-8051-III 型单片机实验箱

三、实验原理

定时常数的确定

定时器/计数器的输入脉冲周期与机器周期一样，为振荡频率的 1/12。本实验中时钟频率为 6.0 MHZ，现要采用中断方法来实现 0.5 秒延时，要在定时器 1 中设置一个时间常数，使其每隔 0.1 秒产生一次中断，CPU 响应中断后将 R0 中计数值减一，令 R0=05H，即可实现 0.5 秒延时。

时间常数可按下述方法确定：

机器周期=12÷晶振频率=12/(6×10⁶)=2us

设计数初值为 X，则 (2e+16-X) ×2×10⁶=0.1，可求得 X=15535

化为十六进制则 X=3CAFH，故初始值为 TH1=3CH，TL1=AFH

初始化程序

包括定时器初始化和中断系统初始化，主要是对 IP、IE、TCON、TMOD 的相应位进行正确的设置，并将时间常数送入定时器中。由于只有定时器中断，IP 便不必设置。

设计中断服务程序和主程序


中断服务程序除了要完成计数减一工作外，还要将时间常数重新送入定时器中，为下一次中断做准备。主程序则用来控制发光二极管按要求顺序燃灭。

由 8031 内部定时器 1 按方式 1 工作，即作为 16 位定时器使用，每 0.1 秒钟 T1 溢出中断一次。P1 口的 P1.0~P1.7 分别接发光二极管的 L1~L8。要求编写程序模拟一循环彩灯。彩灯变化花样可自行设计。例程给出的变化花样为：①L1、L2、...L8 依次点亮；②L1、L2、...L8 依次熄灭；③L1、L2、...L8 全亮、全灭。各时序间隔为 0.5 秒。让发光二极管按以上规律循环显示下去。


1、实验电路图：

2、程序流程图：

主程序框图



中断程序框图



3、实验程序:

```

NAME    T6          ;定时器实验
OUTPORT EQU    0CFB0H
CSEG AT 0000H
LJMP START
CSEG AT 401BH      ;定时器/计数器 1 中断程序入口地址
LJMP INT
CSEG AT 4100H
START:  MOV    A,#01H    ;首显示码
        MOV    R1,#03H   ;:03 是偏移量, 即从基址寄存器到表首的距离
        MOV    R0,#5H    ;:05 是计数值
        MOV    TMOD,#10H ;计数器置为方式 1
  
```

```

MOV    TL1,#0AFH  ;装入时间常数
MOV    TH1,#03CH
ORL    IE,#88H    ;CPU 中断开放标志位和定时器
                    ;1 溢出中断允许位均置位
SETB   TR1        ;开始计数
LOOP1: CJNE   R0,#00,DISP
MOV    R0,#5H     ;R0 计数计完一个周期，重置初值
INC    R1         ;表地址偏移量加 1
CJNE   R1,#31H,LOOP2
MOV    R1,#03H    ;如到表尾，则重置偏移量初值
LOOP2: MOV    A,R1 ;从表中取显示码入累加器
MOVC   A,@A+PC
JMP    DISP
DB     01H,03H,07H,0FH,1FH,3FH,7FH,0FFH,0FEH,0FCH
DB     0F8H,0F0H,0E0H,0C0H,80H,00H,0FFH,00H,0FEH
DB     0FDH,0FBH,0F7H,0EFH,0DFH,0BFH,07FH,0BFH,0DFH
DB     0EFH,0F7H,0FBH,0FDH,0FEH,00H,0FFH,00H
DISP:  MOV    P1,A ;将取得的显示码从 P1 口输出显示
JMP    LOOP1
INT:   CLR    TR1 ;停止计数
DEC    R0       ;计数值减一
MOV    TL1,#0AFH ;重置时间常数初值
MOV    TH1,#03CH
SETB   TR1     ;开始计数
RETI                   ;中断返回
END

```

四、实验步骤

- 1、P1.0~P1.7 分别接发光二极管 L1~L8，运行程序观察发光二极管的点亮情况。
- 2、修改程序，只重复①图案。

五、问题思考

- 1、分析硬件定时和软件定时的区别。
- 2、怎样修改程序可以点亮更多的图案。

实验五 串行口实验

一、实验目的

- (1) 掌握 8031 串行口方式 1 的工作方式及编程方法
- (2) 掌握串行通讯中波特率的设置
- (3) 在给定通讯波特率的情况下，会计算定时时间常数

二、实验仪器

微机、EL-8051-III 型单片机实验箱

三、实验原理

MCS-51 单片机串行通讯的波特率随串行口工作方式选择的不同而不同，它除了与系统的振荡频率 f ，电源控制寄存器 PCON 的 SMOD 位有关外，还与定时器 T1 的设置有关。

- 1) 在工作方式 0 时，波特率固定不变，仅与系统振荡频率有关，其大小为 $f/12$ 。
- 2) 在工作方式 2 时，波特率也只固定为两种情况：

当 SMOD=1 时， 波特率= $f/32$

当 SMOD=0 时， 波特率= $f/64$

- 3) 在工作方式 1 和 3 时，波特率是可变的：

当 SMOD=1 时， 波特率=定时器 T1 的溢出率/16

当 SMOD=0 时， 波特率=定时器 T1 的溢出率/32


其中，定时器 T1 的溢出率= $f/(12*(256-N))$ ，N 为 T1 的定时时间常数。

在实际应用中，往往是给定通讯波特率，而后去确定时间常数。例如： $f=6.144\text{MHZ}$ ，波特率等于 1200，SMOD=0 时，则 $1200=6144000/(12*32*(256-N))$ ，计算得 $N=F2H$ 。


例程中设置串行口工作于方式 1，SMOD=0，波特率为 1200。

循环彩灯的变化花样与实验六相同。也可自行设计变化花样。

1、程序流程图：



2、实验电路图：



3、实验程序：

CSEG AT 0000H

LJMP START

CSEG AT 4100H

PORT EQU 0CFA0H

```

START: MOV TMOD,#20H      ;选择定时器模式 2,计时方式
      MOV TL1,#0F2H      ;预置时间常数, 波特率为 1200
      MOV TH1,#0F2H
      MOV 87H,#00H      ;PCON=00,使 SMOD=0
      SETB TR1          ;启动定时器 1
      MOV SCON           ;#50H;串行口工作于方式 1,允许串行接收
      MOV R1,#12H       ;R1 中存放显示计数值
      MOV DPTR,#TABLE
      MOV A,DPL
      MOV DPTR,#L1
      CLR C
      SUBB A,DPL        ;计算偏移量
      MOV R5,A         ;存放偏移量
      MOV R0,A
SEND:  MOV A,R0
      MOVC A,@A+PC     ;取显示码
      L1: MOV SBUF,A    ;通过串行口发送显示码
      WAIT: JBC R1,L2  ;接收中断标志为 0 时循环等待
    
```

```

        SJMP WAIT
L2: CLR RI          ;接收中断标志清零
        CLR TI      ;发送中断标志清零
        MOV A,SBUF  ;接收数据送 A
        MOV DPTR,#PORT
        MOVX @DPTR,A ;显码输出
        ACALL DELAY ;延时 0.5 秒
        INC R0      ;偏移量下移
        DJNZ R1,SEND ;为零,置计数初值和偏移量初值
        MOV R1,#12H
        MOV A,R5
        MOV R0,A
        JMP SEND

TABLE: DB 01H, 03H, 07H, 0FH, 1FH, 3FH,7FH,0FFH,0FEH
        DB 0FCH,0F8H,0F0H,0E0H,0C0H,80H,00H,0FFH,00H
DELAY: MOV R4,#05H      ;延时 0.5 秒
        DEL1: MOV R3,#200
        DEL2: MOV R2,#126
        DEL3: DJNZ R2,DEL3
                DJNZ R3,DEL2
                DJNZ R4,DEL1
        RET
        END

```

四、实验步骤

- 1、8031 的 TXD 接 RXD；74LS273 的 CS273 接 CS0；O0~O7 接发光二极管的 L1~L8。
- 2、运行程序，观察循环彩灯的点亮情况。
- 3、修改程序，改变彩灯的图案。

五、问题思考

- 1、拔掉串口线，会发生什么现象？怎样解释？
- 2、重新插上串口线，怎样修改程序才能继续正常工作？

实验六 A/D 转换实验

一、实验目的

- (1) 学习 A/D 转换与单片机的接口方法
- (2) 学习了解 A/D 芯片 ADC0809 转换性能及编程方法

二、实验仪器

微机、EL-8051-III 型单片机实验箱

三、实验原理

A/D 转换器大致有三类：一是双积分 A/D 转换器，优点是精度高，抗干扰性好，价格便宜，但速度慢；二是逐次逼近法 A/D 转换器，精度、速度、价格适中；三是并行 A/D 转换器，速度快，价格也昂贵。

实验用的 ADC0809 属第二类，是八位 A/D 转换器。每采集一次需 100us。

ADC0809 START 端为 A/D 转换启动信号，ALE 端为通道选择地址的锁存信号。实验电路中将其相连，以便同时锁存通道地址并开始 A/D 采样转换，故启动 A/D 转换只需如下两条指令：

```
MOV DPTR, #PORT
```

```
MOVX @DPTR,A
```


A 中为何内容并不重要，这是一次虚拟写。

在中断方式下，A/D 转换结束后会自动产生 EOC 信号，将其与 8031CPU 板上的 INT0 相连接。在中断处理程序中，使用如下指令即可读取 A/D 转换的结果：


```
MOV DPTR, #PORT
```

```
MOVX A, @DPTR
```

1、程序流程图：



2、实验电路图：



3、实验程序：

```
PORT EQU 0CFA0H
```

```
    CSEG AT 0000H
```

```
    LJMP START
```

```
    CSEG AT 4100H
```

```
START: MOV DPTR,#PORT      ;启动通道 0
```

```
    MOVX @DPTR,A
```

```
    MOV R0,#0FFH
```

```
LOOP1: DJNZ R0,LOOP1      ;等待中断
```

```
    MOVX A,@DPTR
```

```
    MOV R1,A
```

```
DISP: MOV A,R1            ;从 R1 中取转换结果
```

```
    SWAP A                ;分离高四位和低四位
```

```
    ANL A,#0FH            ;并依次存放在 50H 到 51H 中
```

```
    MOV 50H,A
```

```
    MOV A,R1
```

```
    ANL A,#0FH
```

```
    MOV 51H,A
```

```
LOOP: MOV DPTR,#0CFE9H   ;写显示 RAM 命令字
```

```
    MOV A,#90H
```

```
    MOVX @DPTR,A
```

```
    MOV R0,#50H          ;存放转换结果地址初值送 R0
```

```
    MOV R1,#02H
```

```

MOV DPTR,#0CFE8H      ;8279 数据口地址
DL0: MOV A,@R0
ACALL TABLE          ;转换为显码
MOVX @DPTR,A          ;送显码输出
INC R0
DJNZ R1,DL0
SJMP DEL1
TABLE: INC A
MOVC A,@A+PC
RET
DB 3FH,06H,5BH,4FH,66H,6DH,7DH,07H
DB 7FH,6FH,77H,7CH,39H,5EH,79H,71H
DEL1: MOV R6,#255      ;延时一段时间使显示更稳定
DEL2: MOV R5,#255
DEL3: DJNZ R5,DEL3
      DJNZ R6,DEL2
      LJMP START      ;循环
END

```

四、实验步骤

- 1、0809 的片选信号 CS0809 接 CS0, 电位器的输出信号 AN0 接 0809 的 ADIN0, EOC 接 CPU 板的 INTO。
- 2、运行程序, 观察数码管的显示情况。
- 3、调节电位器的位置, 观察数码管的变化情况。

五、问题思考

- 1、分析电位器在电路中的作用。
- 2、ADC0809 与单片机是怎样联络的?